



HAL
open science

Axiomes de continuité en géométrie neutre : une étude mécanisée en Coq

Charly Gries, Julien Narboux, Pierre Boutry

► To cite this version:

Charly Gries, Julien Narboux, Pierre Boutry. Axiomes de continuité en géométrie neutre : une étude mécanisée en Coq. 2018. hal-01923814v1

HAL Id: hal-01923814

<https://hal.science/hal-01923814v1>

Preprint submitted on 15 Nov 2018 (v1), last revised 4 Jan 2019 (v2)

HAL is a multi-disciplinary open access archive for the deposit and dissemination of scientific research documents, whether they are published or not. The documents may come from teaching and research institutions in France or abroad, or from public or private research centers.

L'archive ouverte pluridisciplinaire **HAL**, est destinée au dépôt et à la diffusion de documents scientifiques de niveau recherche, publiés ou non, émanant des établissements d'enseignement et de recherche français ou étrangers, des laboratoires publics ou privés.

Axiomes de continuité en géométrie neutre : une étude mécanisée en Coq

Charly Gries, Julien Narboux, and Pierre Boutry

Laboratoire ICube, UMR 7357 CNRS, Université de Strasbourg
charly.gries@etu.unistra.fr, {narboux, boutry}@unistra.fr

Abstract

Dans cet article, nous présentons les résultats sur la continuité que nous avons obtenus dans le cadre du projet *GeoCoq*, qui constitue une formalisation en Coq de la géométrie : nous avons défini 11 axiomes de continuité et établi leur hiérarchie. Nos démonstrations ont été formalisées dans le cadre de la géométrie neutre de Tarski en dimension quelconque et de la logique intuitionniste.

Introduction

Les commentateurs des *Éléments* d’Euclide ont remarqué depuis longtemps que certaines démonstrations d’Euclide ne découlent pas formellement de ses postulats, car Euclide admet implicitement l’existence de certains points. C’est le cas par exemple dans la première proposition du livre I : celle-ci construit un triangle équilatéral sur un segment donné en utilisant l’intersection de deux cercles, sans que l’existence d’une telle intersection n’ait été prouvée ou postulée.

Dans cet article nous dénotons par le mot continuité des axiomes qui affirment l’existence de points sous certaines conditions, par exemple l’existence de l’intersection de deux cercles intriqués. Nous nous concentrons sur des propriétés géométriques, bien que des propriétés algébriques analogues existent.

En 1882, Moritz Pasch a proposé un axiome qui permet de combler un autre type de lacune dans les *Éléments* [Pas82]. L’axiome de Pasch consiste à supposer que quand une droite pénètre à l’intérieur d’un triangle par un côté, elle ressort par un des deux autres côtés. On peut considérer cette propriété comme un axiome de continuité concernant l’intersection des droites ; cependant, l’axiome de Pasch ne fait pas partie des axiomes étudiés dans cet article, car il appartient au contexte de notre étude.

Au tournant du XX^{ème} siècle, David Hilbert puis Alfred Tarski ont proposé des développements systématiques de la géométrie sans supposer d’axiome concernant l’intersection de deux cercles [Hil77, SST83]. En effet, une grande part de la géométrie euclidienne peut être développée sans axiome de continuité (l’axiome de Pasch mis à part) : il est possible de construire des coordonnées et de prouver la plupart des théorèmes de géométrie [BBN18]. Hilbert et Tarski introduisent tout de même des axiomes de continuité dans un objectif principalement métathéorique.

Notre bibliothèque *GeoCoq* contient une formalisation des fondements de la géométrie basée sur les axiomatiques d’Euclide, Hilbert et Tarski. Nous avons prouvé précédemment que les axiomes de Hilbert des groupes I, II et III sont mutuellement interprétables avec les axiomes A_1 – A_8 de Tarski [BN12, BBN16]. Dans un premier temps, notre formalisation des systèmes d’axiomes de Tarski et de Hilbert n’incluait pas la continuité. En géométrie euclidienne, la continuité n’est nécessaire que pour certaines constructions géométriques et pour définir les fonctions trigonométriques inverses.

Dans le cadre de la géométrie neutre¹, la continuité joue également un rôle dans la classification des différentes versions du postulat des parallèles [BGNS17] : certaines sont équivalentes entre elles dans

1. La géométrie neutre recouvre l’ensemble des résultats communs entre la géométrie euclidienne et la géométrie hyperbolique. C’est-à-dire qu’il existe des droites parallèles mais elles ne sont pas forcément uniques. Cela exclut la géométrie sphérique.

toute géométrie neutre, comme l'unicité de la parallèle et la transitivité du parallélisme ; d'autres ne le sont que sous l'hypothèse d'axiomes de continuité plus ou moins forts. Ainsi, le postulat du triangle, qui stipule que la somme des angles d'un triangle est égale à π (ou plutôt à deux droits), est une conséquence de l'unicité de la parallèle, mais la réciproque n'est pas vraie : Max Dehn a démontré l'existence d'un modèle non-archimédien dans lequel la somme des angles d'un triangle vaut π mais l'unicité de la parallèle n'est pas vérifiée [Deh00]. Cependant, sous l'hypothèse de l'axiome d'Archimède, les deux postulats sont bel et bien équivalents.

Dans cet article, nous étendons notre étude des liens entre les axiomatique de Hilbert et de Tarski aux propriétés de continuité. Nos preuves formelles reposent sur le développement systématique de la géométrie que Wolfram Schwabhäuser, Wanda Szmielew et Alfred Tarski ont produit à partir du système de Tarski [SST83]. Elles s'inspirent des travaux de Marvin J. Greenberg [Gre10] et de George E. Martin [Mar98], ainsi que de l'ouvrage de Franz Rothe [Rot14]. Cependant, les preuves de ces écrits ne sont parfois qu'esquissées, et elles ne sont pas vérifiées mécaniquement, ce qui permet d'échapper à la démonstration parfois laborieuse de résultats intermédiaires semblant évidents, notamment concernant la non-dégénérescence des figures et la position relative des points. De plus, comme nous nous restreignons la plupart du temps à la logique intuitionniste, certaines preuves utilisant la loi du tiers exclu ont dû être adaptées pour ce contexte.

Cet article a pour objet de faire le bilan des propriétés de continuité que nous avons étudiées. Nous allons présenter dans la partie 1 les axiomes de continuité que nous avons formalisés, avant de présenter leur hiérarchie et expliquer quelques démonstrations dans la partie 2.

1 Formalisation des axiomes de continuité

Nos preuves sont données dans le cadre des axiomes A_1 – A_8 du système de Tarski pour la géométrie neutre en dimension quelconque² [SST83], que nous ne détaillons pas ici. Nous avons décrit ces axiomes et leur formalisation en français dans [GBN16] et de manière plus étendue dans [BGNS17]. Nous avons présenté dans ces articles plusieurs axiomes de continuité, dont ceux que nous appelons axiome d'Aristote et axiome de Greenberg.

Rappelons que le système de Tarski est basé sur un seul type primitif, représentant les points, et deux prédicats, à savoir la congruence (ou équidistance) et l'interposition (en anglais *betweenness*) :

- Cong $ABCD$ indique que les segments AB et CD sont de même longueur ;
- Bet ABC signifie que A , B et C sont colinéaires et que B est entre A et C .

Ces deux prédicats permettent notamment de définir les prédicats suivants de [SST83] :

- Col ABC : A , B et C sont colinéaires ;
- Midpoint MAB : M est le milieu du segment AC ;
- Out PAB : les points A et B se situent sur la même demi-droite ouverte d'origine P ;
- Le $ABCD$: la longueur AB est inférieure ou égale à la longueur CD ;
- Lt $ABCD$: la longueur AB est strictement inférieure à la longueur CD .

Nous travaillons principalement en logique intuitionniste ; toutefois, nous nous autorisons à raisonner par tiers exclu sur l'égalité de points.

1.1 Axiomes de continuité de Tarski

Considérons à présent les axiomes de continuité utilisés par Tarski dans son article fondateur [Tar51]. Nous en exposerons d'abord la présentation donnée par Tarski, avant de développer leur signification et expliquer notre formalisation.

2. Nous ne considérons que les espaces de dimension supérieure ou égale à 2.

1.1.1 Axiome de Dedekind

L'axiome de continuité de Dedekind est un axiome du deuxième ordre³ : il comprend une quantification sur des ensembles α et β .

Axiome de Dedekind A_{11}

$$\forall \alpha \beta (\exists a \forall xy (x \in \alpha \wedge y \in \beta \Rightarrow \text{Bet } axy)) \Rightarrow (\exists b \forall xy (x \in \alpha \wedge y \in \beta \Rightarrow \text{Bet } xby))$$

La formalisation de cet axiome en Coq est directe ; nous quantifions sur les ensembles α et β formalisés par des prédicats sur les points :

```
Definition dedekind_s_axiom := forall (Alpha Beta : Tpoint -> Prop),
  (exists A, forall X Y, Alpha X -> Beta Y -> Bet A X Y) ->
  (exists B, forall X Y, Alpha X -> Beta Y -> Bet X B Y).
```




FIGURE 1 – Axiome de Dedekind

On peut déduire de cet axiome l'ensemble des axiomes de continuité. Sans recourir à la quantification sur les ensembles, qui ne sont pas définissables dans le système de Tarski, il exprime la résolution des coupures de Dedekind (fig. 1⁴), qui caractérise les espaces complets.

En effet, sous les hypothèses de cet axiome, les points X vérifiant $\text{Alpha } X$ et les points Y vérifiant $\text{Beta } Y$ sont situés sur une même demi-droite d'origine A ⁵, et on peut ordonner totalement cette demi-droite en utilisant la relation d'interposition et le point A . Munis de cet ordre, nous pouvons dire que tout point X est inférieur à tout point Y . Si de plus les points X et Y forment une partition de la demi-droite (c'est-à-dire qu'ils décrivent l'ensemble de la demi-droite sans se rencontrer), on dit qu'ils forment une coupure de Dedekind de cette demi-droite. Construire un point B situé entre tout point X et tout point Y revient alors à exhiber la borne inférieure de l'ensemble des points Y , c'est-à-dire à résoudre la coupure de Dedekind associée.

Sur la figure 1, les points X vérifiant $\text{Alpha } X$ se situent au sein du trait plein, tandis que les points Y vérifiant $\text{Beta } Y$ se trouvent parmi le trait pointillé.

Pour les besoins de notre étude, nous avons introduit une variante plus explicite de cet axiome :

3. Les formules du premier ordre sont des formules dans lesquelles on ne peut quantifier que sur les variables, c'est-à-dire, dans ce contexte, sur les points.

4. Nous présenterons la plupart des figures à la fois dans le modèle euclidien et dans un modèle non-euclidien : le disque de Poincaré. La figure de gauche illustrera la validité de l'axiome en géométrie euclidienne. La figure de droite jouera un rôle analogue dans le disque de Poincaré.

5. À condition qu'il existe un tel point X distinct de A .

```

Definition dedekind_variant := forall (Alpha Beta : Tpoint -> Prop) A C,
  Alpha A -> Beta C -> (forall P, Out A P C -> Alpha P \/ Beta P) ->
  (forall X Y, Alpha X -> Beta Y -> Bet A X Y /\ X <> Y) ->
  (exists B, forall X Y, Alpha X -> Beta Y -> Bet X B Y).

```

Dans cette version, les points X vérifiant $\text{Alpha } X$ et les points Y vérifiant $\text{Beta } Y$ forment nécessairement une coupure de Dedekind de la demi-droite $[AC$. Cette précision fait de cette version une propriété plus facile à prouver; les deux formulations sont toutefois équivalentes en logique classique.

1.1.2 Schéma d'axiomes de Dedekind de premier ordre

La propriété de continuité A'_{11} introduite par Tarski est un schéma d'axiomes de premier ordre qui restreint la résolution des coupures de Dedekind à celles qui sont définissables au premier ordre :

Schéma de Dedekind de premier ordre A'_{11}

$$(\exists a \forall xy (\alpha(x) \wedge \beta(y) \Rightarrow \text{Bet } axy)) \Rightarrow (\exists b \forall xy (\alpha(x) \wedge \beta(y) \Rightarrow \text{Bet } xby))$$

où $\alpha(x)$ (respectivement $\beta(y)$) représente n'importe quelle formule qui ne contienne aucune occurrence libre de a, b, y (resp. a, b, x) et qui soit du premier ordre dans le langage du système d'axiomes de Tarski.

Pour formaliser ce schéma d'axiomes dans la logique d'ordre supérieur de Coq, nous devons transformer ce schéma en un axiome du deuxième ordre tout en restreignant la quantification sur $\alpha(x)$ et $\beta(y)$ aux formules définissables au premier ordre dans le langage du système d'axiomes de Tarski.

Nous utilisons la définition inductive suivante, qui définit un prédicat permettant d'indiquer si une formule est de premier ordre (en anglais *first-order formula*).

```

Inductive FOF : Prop -> Prop :=
| eq_fof : forall A B:Tpoint, FOF (A = B)
| bet_fof : forall A B C, FOF (Bet A B C)
| cong_fof : forall A B C D, FOF (Cong A B C D)
| not_fof : forall P, FOF P -> FOF (~ P)
| and_fof : forall P Q, FOF P -> FOF Q -> FOF (P /\ Q)
| or_fof : forall P Q, FOF P -> FOF Q -> FOF (P \/ Q)
| implies_fof : forall P Q, FOF P -> FOF Q -> FOF (P -> Q)
| forall_fof : forall P, (forall (A:Tpoint), FOF (P A)) -> FOF (forall A, P A)
| exists_fof : forall P, (forall (A:Tpoint), FOF (P A)) -> FOF (exists A, P A).

```

Afin de nous convaincre que cette définition est correcte, nous avons proposé une définition alternative : nous disons qu'une formule est de premier ordre si elle est l'interprétation d'un terme dans un langage de la théorie du premier ordre. Nous avons d'abord formalisé ce langage :

```

Inductive tFOF :=
| eq_fof1 : Tpoint -> Tpoint -> tFOF
| bet_fof1 : Tpoint -> Tpoint -> Tpoint -> tFOF
| cong_fof1 : Tpoint -> Tpoint -> Tpoint -> Tpoint -> tFOF
| not_fof1 : tFOF -> tFOF
| and_fof1 : tFOF -> tFOF -> tFOF
| or_fof1 : tFOF -> tFOF -> tFOF
| implies_fof1 : tFOF -> tFOF -> tFOF
| forall_fof1 : (Tpoint -> tFOF) -> tFOF
| exists_fof1 : (Tpoint -> tFOF) -> tFOF.

```

Nous avons ensuite défini l'interprétation de notre langage dans les propositions de Coq, par récurrence sur la structure des termes :

```
Fixpoint fofl_prop (F:tFOF) := match F with
| eq_fofl A B => A = B
| bet_fofl A B C => Bet A B C
| cong_fofl A B C D => Cong A B C D
| not_fofl F1 => ~ fofl_prop F1
| and_fofl F1 F2 => fofl_prop F1 /\ fofl_prop F2
| or_fofl F1 F2 => fofl_prop F1 \/ fofl_prop F2
| implies_fofl F1 F2 => fofl_prop F1 -> fofl_prop F2
| forall_fofl P => forall A, fofl_prop (P A)
| exists_fofl P => exists A, fofl_prop (P A) end.
```

Enfin, nous avons prouvé que les deux définitions sont équivalentes dans le sens suivant :

- l'interprétation de tout terme du premier ordre est une formule du premier ordre ;
- pour toute formule du premier ordre, il existe un terme du premier ordre dont l'interprétation est équivalente à la formule⁶.

```
Lemma fofl__fof : forall F1:tFOF, FOF (fofl_prop F1).
```

```
Lemma fof__fofl : FunctionalChoice_on Tpoint tFOF ->
forall F:Prop, FOF F -> exists F1:tFOF, F <-> fofl_prop F1.
```

Cela justifie notre définition formelle du schéma d'axiomes de continuité :

```
Definition first_order_dedekind := forall Alpha Beta,
(forall X, FOF (Alpha X)) -> (forall Y, FOF (Beta Y)) ->
(exists A, forall X Y, Alpha X -> Beta Y -> Bet A X Y) ->
(exists B, forall X Y, Alpha X -> Beta Y -> Bet X B Y).
```

1.2 Axiomes de continuité de Hilbert

Dans cette partie, nous allons décrire notre formalisation des axiomes de continuité de Hilbert tels qu'ils sont présentés dans les *Fondements de la Géométrie* [Hi177].

1.2.1 Axiome d'Archimède

Le premier axiome de continuité utilisé par Hilbert est l'axiome d'Archimède (fig. 2).

V-1 : Axiome d'Archimède Si AB et CD sont deux segments quelconques, il existe un nombre entier n tel que le report du segment AB répété n fois à partir de C sur la demi-droite déterminée par D , conduit à un point situé au-delà de D ⁷.

En d'autres termes, étant donnés deux segments AB et CD avec A distinct de B , il existe un entier n et $n + 1$ points A_0, \dots, A_n de la droite CD , tels que A_j se situe entre A_{j-1} et A_{j+1} pour $0 < j < n$, $A_j A_{j+1}$ et AB ont même longueur pour $0 \leq j < n$, $A_0 = C$ et D se situe entre A_0 et A_n . Sur la figure 2, l'entier n est égal à 4.

6. La preuve de cette affirmation repose sur l'hypothèse `FunctionalChoice`, qui établit qu'une fonction peut être construite à partir d'une relation totale à gauche. Le reste du développement Coq ne repose pas sur cette hypothèse.

7. Par souci de cohérence avec notre développement Coq, nous avons interverti les rôles de AB et CD par rapport à la formulation de Hilbert dans [Hi177].




FIGURE 2 – Axiome d'Archimède

On dira qu'un espace géométrique est archimédien s'il vérifie l'axiome d'Archimède. Cet axiome peut être dérivé de A_{11} , mais pas de A'_{11} , car il s'agit d'une propriété du deuxième ordre.

En nous inspirant du travail de Jean Duprat [Dup10], nous avons défini l'axiome d'Archimède de façon inductive, sans introduire explicitement les entiers naturels. Pour cela, nous avons d'abord traduit le fait "il existe un entier n et $n + 1$ points A_0, \dots, A_n de la droite AB , tels que A_j se situe entre A_{j-1} et A_{j+1} pour $0 < j < n$, $A_j A_{j+1}$ et AB ont même longueur pour $0 \leq j < n$, $A_0 = A$ et $A_n = C$ " par le prédicat $\text{Grad} : \text{Grad } A \ B \ C$ signifie que C appartient à la demi-droite $[AB$ et que la distance AC est un multiple de la distance AB .

```

Inductive Grad : Tpoint -> Tpoint -> Tpoint -> Prop :=
| grad_init : forall A B, Grad A B B
| grad_stab : forall A B C C',
  Grad A B C ->
  Bet A C C' -> Cong A B C C' ->
  Grad A B C'.

```

```

Definition Reach A B C D := exists B', Grad A B B' /\ Le C D A B'.

```

$\text{Grad } A \ B \ B'$ indique que B' fait partie de la graduation basée sur le segment AB . La notation $\text{Reach } A \ B \ C \ D$ signale donc que le segment CD est inférieur ou égal à un segment de la forme $A_0 A_n$ basé sur le segment AB . Les propriétés de la comparaison de segments permettent d'en dériver une définition de l'axiome d'Archimède.

```

Definition archimedes_axiom := forall A B C D, A <> B -> Reach A B C D.

```

Notons que nous avons précédemment étudié une propriété analogue portant sur les angles, à savoir que tout angle aigu non nul peut être ajouté à lui-même jusqu'à obtenir un angle obtus. En nous inspirant de la démonstration présentée par Hartshorne [Har00], nous avons pu établir que cette propriété était une conséquence de l'axiome d'Archimède. Cependant, comme nous n'avons pas pu la relier à un autre axiome de continuité, nous n'en donnerons pas ici une définition formelle, qui nécessiterait d'introduire les notions nécessaires à la somme d'angles⁸, ainsi que la variante du prédicat Grad associée.

8. Nous avons précédemment présenté en français notre définition formelle de la somme d'angles [GBN16].

1.2.2 Axiomes d'intégrité de Hilbert

Dans la première édition des *Fondements de la géométrie* [Hil99], l'axiome d'Archimède est le seul axiome de continuité. De la deuxième édition à la sixième édition, il introduit également l'axiome d'intégrité, qui n'est pas utilisé dans le développement si ce n'est pour des considérations métamathématiques. Dans la septième édition, suite à une remarque de Paul Bernays, l'axiome d'intégrité est remplacé par l'axiome d'intégrité linéaire, qui implique l'axiome d'intégrité.

L'axiome d'intégrité peut paraître déroutant au premier abord. En effet, cet axiome est une propriété à propos des modèles des autres axiomes :

Axiome d'intégrité Les éléments de la géométrie (les points, les droites et les plans) constituent un ensemble qui n'est susceptible d'aucune extension si les axiomes d'appartenance, d'ordre, de congruence et l'axiome d'Archimède sont conservés.

Nous avons interprété cette définition en considérant qu'un modèle \mathcal{M} des axiomes de la géométrie respecte l'axiome d'intégrité s'il n'existe aucune fonction f de \mathcal{M} dans un modèle archimédien \mathcal{M}' vérifiant les trois propriétés suivantes :

- f préserve les propriétés géométriques (ici l'interposition et la congruence);
- f est injective (f envoie deux points distincts sur deux images distinctes);
- f n'est pas surjective (certains points de \mathcal{M}' sont inaccessibles par f).

Une autre façon de le formuler est que sous l'axiome d'intégrité, toute fonction injective vers un modèle archimédien et préservant les propriétés géométriques est également surjective, ce qui fait de f un isomorphisme. Ainsi, les isométries sont des isomorphismes. On peut considérer les isomorphismes comme des extensions non propres, et qualifier de propres les extensions non surjectives.

Remarquons que notre formalisation de l'axiome fait référence aux axiomes A_1 – A_8 de Tarski en lieu et place des groupes d'axiomes d'appartenance (I), d'ordre (II) et de congruence (III) de Hilbert. Ce choix est motivé par souci de cohérence avec le reste de la bibliothèque *GeoCoq* mais n'a pas d'impact, puisque nous avons déjà formalisé la preuve que les systèmes d'axiomes en question sont mutuellement interprétables [BN12, BBN16].

Eduardo Giovannini a établi que Hilbert connaissait l'axiome de continuité de Dedekind A_{11} et a choisi l'axiome d'intégrité parce qu'il n'implique pas l'axiome d'Archimède, tandis que la conjonction de l'axiome d'intégrité et de celui d'Archimède caractérise les espaces complets [Gio13]. En revanche, nous avons pu formaliser la preuve que la négation de l'axiome d'Archimède implique l'axiome d'intégrité : un modèle non-archimédien ne peut disposer d'une extension (propre ou non propre) vers un modèle archimédien. À l'opposé, tout modèle (archimédien ou non) peut être muni d'une extension propre vers un modèle non-archimédien [Hil77]; omettre l'axiome d'Archimède dans l'énoncé de l'axiome d'intégrité aboutit donc à un énoncé contradictoire. Le rôle de cet axiome et sa relation avec celui d'Archimède sont discutés dans [Ehr97].

Dans [Hil77], les espaces géométriques sont supposés être de dimension 3, même s'il est possible de modifier les axiomes pour travailler sur le plan ou en dimension quelconque. Cette formulation de l'axiome d'intégrité suppose donc implicitement que les deux modèles concernés sont de même dimension finie (en l'occurrence 3). Cette propriété est fondamentale pour l'intégrité de cet axiome : si on oublie cette hypothèse, on obtient un axiome contradictoire avec les espaces archimédiens. Ainsi, tout espace archimédien \mathcal{M}' de dimension 3 dispose d'un infini de sous-espaces \mathcal{M} de dimension 2, pour lesquels l'inclusion de \mathcal{M} dans \mathcal{M}' est une extension propre. Ne disposant pas d'un prédicat pour exprimer le fait pour deux espaces d'être de même dimension finie, nous avons formalisé deux versions de cette axiome convenant respectivement aux espaces de dimension 2 et 3 : elles consistent à affirmer que toute extension vers un espace respectivement de dimension 2 ou 3 est non propre⁹.

9. Dans notre développement, Tn désigne le contexte courant, à savoir les axiomes de Tarski pour la géométrie neutre.

```

Definition inj {T1 T2:Type} (f:T1->T2) := forall A B, f A = f B ->
  A = B.

Definition pres_bet {Tm: Tarski_neutral_dimensionless}
  (f : @Tpoint Tn -> @Tpoint Tm) := forall A B C, Bet A B C ->
  Bet (f A) (f B) (f C).

Definition pres_cong {Tm: Tarski_neutral_dimensionless}
  (f : @Tpoint Tn -> @Tpoint Tm) := forall A B C D, Cong A B C D ->
  Cong (f A) (f B) (f C) (f D).

Definition extension {Tm: Tarski_neutral_dimensionless}
  (f : @Tpoint Tn -> @Tpoint Tm) := inj f /\ pres_bet f /\ pres_cong f.

Definition completeness_for_planes := forall
  (Tm: Tarski_neutral_dimensionless)
  (Tm2 : Tarski_neutral_dimensionless_with_decidable_point_equality Tm)
  (M : Tarski_2D Tm2)
  (f : @Tpoint Tn -> @Tpoint Tm),
  @archimedes_axiom Tm ->
  extension f ->
  forall A, exists B, f B = A.

Definition completeness_for_3d_spaces := forall
  (Tm: Tarski_neutral_dimensionless)
  (Tm2 : Tarski_neutral_dimensionless_with_decidable_point_equality Tm)
  (M : Tarski_3D Tm2)
  (f : @Tpoint Tn -> @Tpoint Tm),
  @archimedes_axiom Tm ->
  extension f ->
  forall A, exists B, f B = A.

```

Considérons maintenant l'axiome d'intégrité linéaire :

V-2 : Axiome d'intégrité linéaire L'ensemble des points d'une droite, soumis aux relations d'ordre et de congruence, n'est susceptible d'aucune extension dans laquelle sont valables les relations précédentes et les propriétés fondamentales d'ordre linéaire et de congruence déduites des axiomes I–III et de l'axiome V-1.

Remarquons que dans la septième édition des *Fondements de la Géométrie*, Hilbert donne une variante de cet axiome avec une liste plus restreinte des propriétés fondamentales valables dans l'extension. Werner Weber a montré que cette variante était contradictoire [Web38]. L'usage de la preuve formelle peut permettre d'éviter ce genre de désagrément.

Ce nouvel axiome présente l'avantage de ne nécessiter aucune hypothèse sur la dimension des espaces géométriques. En effet, il ne s'agit plus d'étudier les extensions de l'espace \mathcal{M} lui-même, mais des droites de \mathcal{M} ; pour satisfaire l'axiome d'intégrité linéaire, il suffit alors de vérifier que toute extension f d'une droite quelconque PQ couvre l'ensemble de la droite $f(P)f(Q)$, ce qui ne dépend pas de la dimension de l'espace \mathcal{M}' dans lequel elle se trouve.

Nous avons donc pu formaliser cet axiome sans peine, après avoir défini le fait pour f d'être une extension de la droite PQ :

```

Definition inj_line {T:Type} (f:Tpoint->T) P Q := forall A B,
  Col P Q A -> Col P Q B -> f A = f B -> A = B.

Definition pres_bet_line {Tm: Tarski_neutral_dimensionless}
  (f : @Tpoint Tn -> @Tpoint Tm) P Q := forall A B C,

```

```

Col P Q A -> Col P Q B -> Col P Q C ->
Bet A B C -> Bet (f A) (f B) (f C).

Definition pres_cong_line {Tm: Tarski_neutral_dimensionless}
(f : @Tpoint Tn -> @Tpoint Tm) P Q := forall A B C D,
Col P Q A -> Col P Q B -> Col P Q C -> Col P Q D ->
Cong A B C D -> Cong (f A) (f B) (f C) (f D).

Definition line_extension {Tm: Tarski_neutral_dimensionless} f P Q :=
P <> Q /\ inj_line f P Q /\ pres_bet_line f P Q /\ pres_cong_line f P Q.

Definition line_completeness := forall
(Tm: Tarski_neutral_dimensionless)
(Tm2 : Tarski_neutral_dimensionless_with_decidable_point_equality Tm)
P Q (f : @Tpoint Tn -> @Tpoint Tm),
@archimedes_axiom Tm ->
line_extension f P Q ->
forall A, Col (f P) (f Q) A -> exists B, Col P Q B /\ f B = A.

```

La démonstration de Paul Bernays prouve que l’axiome d’intégrité linéaire implique l’axiome d’intégrité dans les espaces de dimension 3. En plus de la formaliser, nous avons pu l’adapter pour les espaces de dimension 2.

1.3 Axiome des segments emboîtés de Cantor

Georg Cantor a introduit la propriété des segments emboîtés dans sa première preuve que l’ensemble des points d’un segment est indénombrable [Can74].

Axiome des segments emboîtés Étant donnée une suite $([A_n B_n])_{n \in \mathbb{N}}$ de segments fermés emboîtés, il existe un point X appartenant à tous les segments $A_n B_n$.




FIGURE 3 – Axiome des segments emboîtés de Cantor

Les segments $A_n B_n$ sont dits emboîtés si pour tout entier n , le segment $A_n B_n$ contient le segment $A_{n+1} B_{n+1}$; par commodité, il est d’usage de considérer que les points A_n , A_{n+1} , B_{n+1} et B_n sont alignés dans cet ordre (fig. 3).

Là encore, nous avons affaire à un axiome du deuxième ordre, car il contient une quantification sur des suites de points.

Pour décrire la suite des extrémités des segments, nous allons utiliser des prédicats A et B prenant en argument un entier n et un point P : $A \ n \ P$ (respectivement $B \ n \ P$) signifie que P correspond au point A_n (resp. B_n). Dans cette optique, $\text{Nested } A \ B$ indique que A et B jouent ce rôle pour des segments emboîtés :

```

Definition Nested (A B:nat -> Tpoint -> Prop) :=
(forall n, exists An Bn, A n An /\ B n Bn) /\
forall n An Am Bm Bn,
A n An -> A (S n) Am -> B (S n) Bm -> B n Bn ->
Bet An Am Bm /\ Bet Am Bm Bn /\ Am <> Bm.

Definition cantor_s_axiom := forall A B, Nested A B ->
exists X, forall n An Bn, A n An -> B n Bn -> Bet An X Bn.

```

On peut remarquer que notre définition ne suppose pas que pour tout entier n , les points A_n et B_n soient définis de façon unique. Le contraire aurait alourdi la formalisation sans jouer un rôle crucial, ce que nous avons vérifié formellement. De plus, nous supposons que les points A_n et B_n sont toujours différents entre eux, ce qui ne modifie pas foncièrement l'axiome mais facilite son usage.

1.4 Axiomes de continuité du cercle

Les axiomes de continuité du cercle, ajoutés aux axiomes de la géométrie, décrivent ce qu'on appelle la géométrie à la règle et au compas, pratiquée déjà par Euclide. On en distingue souvent trois (fig. 4) :

Continuité cercle-cercle Étant donnés deux cercles C_1 et C_2 , si deux points de C_1 sont respectivement à l'intérieur et à l'extérieur de C_2 , alors il existe un point d'intersection des deux cercles.

Continuité cercle-droite Étant donnée une droite l contenant un point à l'intérieur du cercle C , il existe un point de l qui se trouve également sur C .

Continuité cercle-segment Étant donné un segment PQ et un cercle C avec P à l'intérieur et Q à l'extérieur de C , il existe un point du segment PQ qui se trouve également sur C .




FIGURE 4 – Continuité du cercle

Pour formaliser des axiomes en Coq, nous avons besoin des prédicats suivants, qui permettent respectivement d'affirmer qu'un point P est sur, à l'intérieur, à l'extérieur, strictement à l'intérieur ou strictement à l'extérieur d'un cercle de centre A et de rayon AB .

```

Definition OnCircle P A B := Cong A P A B.
Definition InCircle P A B := Le A P A B.
Definition OutCircle P A B := Le A B A P.
Definition InCircleS P A B := Lt A P A B.
Definition OutCircleS P A B := Lt A B A P.

```

À présent, nous pouvons formaliser les axiomes et leurs variantes. Commençons par l'axiome de continuité cercle-cercle. Sa variante `circle_circle_two` affirme qu'il existe deux points d'intersection. Nous avons prouvé qu'elle est équivalente à la première formulation, le deuxième point d'intersection pouvant être construit à partir du premier. Ce n'est pas complètement trivial : pour procéder ainsi, il est nécessaire de prouver qu'il est possible de prendre l'image d'un point par la réflexion par rapport à une droite sans utiliser aucun axiome de continuité. Les preuves requises ont été développées par Haragauri Nayanar Gupta [Gup65], reprises dans [SST83] et formalisées précédemment [BN12].

```

Definition circle_circle := forall A B C D P Q,
  OnCircle P C D -> OnCircle Q C D ->
  InCircle P A B -> OutCircle Q A B ->
  exists Z, OnCircle Z A B /\ OnCircle Z C D.

Definition circle_circle_two := forall A B C D P Q,
  OnCircle P C D -> OnCircle Q C D ->
  InCircle P A B -> OutCircle Q A B ->
  exists Z1 Z2,
    OnCircle Z1 A B /\ OnCircle Z1 C D /\
    OnCircle Z2 A B /\ OnCircle Z2 C D /\
    (InCircleS P A B -> OutCircleS Q A B -> Z1<>Z2).

```

Une deuxième variante équivalente présente le double avantage d'être symétrique et d'être exactement la propriété requise pour prouver la première proposition des *Éléments* d'Euclide :

```

Definition circle_circle_bis := forall A B C D P Q,
  OnCircle P C D ->
  InCircle P A B ->
  OnCircle Q A B ->
  InCircle Q C D ->
  exists Z, OnCircle Z A B /\ OnCircle Z C D.

```

Une autre proposition du premier livre des *Éléments* d'Euclide est quant à elle équivalente à l'axiome de continuité cercle-cercle :

Proposition 22 Avec trois droites égales à trois droites données construire un triangle; il faut que deux de ces trois droites, de quelque manière qu'elles soient prises, soient plus grandes que la troisième.

Ici, le mot "droites" désigne en fait des longueurs, et prendre deux longueurs signifie considérer leur somme. En d'autres termes, cette proposition stipule qu'un triangle peut être construit à partir de trois longueurs données à condition que les inégalités triangulaires associées soient respectées. Pour la définir, nous utilisons un prédicat décrivant la somme de longueurs : $\text{SumS } A B C D E F$ exprime le fait que la longueur du segment EF est égale à la somme des longueurs des segments AB et CD .

```

Definition SumS A B C D E F := exists P Q R,
  Bet P Q R /\ Cong P Q A B /\ Cong Q R C D /\ Cong P R E F.

Definition euclid_s_prop_1_22 := forall A B C D E F A' B' C' D' E' F',
  SumS A B C D E' F' -> SumS A B E F C' D' -> SumS C D E F A' B' ->
  Le E F E' F' -> Le C D C' D' -> Le A B A' B' ->
  exists P Q R, Cong P Q A B /\ Cong P R C D /\ Cong Q R E F.

```

En nous inspirant de `circle_circle_bis`, nous avons défini une autre variante équivalente utilisant uniquement les prédicats de base du système de Tarski, afin de faciliter la définition d'une *type class* décrivant le contexte de la géométrie à la règle et au compas :

```

Definition circle_circle_axiom := forall A B C D B' D',
  Cong A B' A B -> Cong C D' C D ->
  Bet A D' B -> Bet C B' D ->
  exists Z, Cong A Z A B /\ Cong C Z C D.

```

Définissons à présent deux versions de l’axiome de continuité cercle-droite affirmant respectivement l’existence d’un ou deux points d’intersection. L’équivalence entre les deux formulations se démontre de façon analogue à l’équivalence entre les deux premières versions de l’axiome de continuité cercle-cercle.

```

Definition one_point_line_circle := forall A B U V P,
  Col U V P -> U <> V -> Bet A P B ->
  exists Z, Col U V Z /\ OnCircle Z A B.
    
```

```

Definition two_points_line_circle := forall A B U V P,
  Col U V P -> U <> V -> Bet A P B ->
  exists Z1 Z2, Col U V Z1 /\ OnCircle Z1 A B /\
    Col U V Z2 /\ OnCircle Z2 A B /\
    Bet Z1 P Z2 /\ (P <> B -> Z1 <> Z2).
    
```

Enfin, la propriété cercle-segment affirme que si un segment a ses extrémités respectivement à l’intérieur et à l’extérieur du cercle, alors il traverse le cercle.

```

Definition segment_circle := forall A B P Q,
  InCircle P A B ->
  OutCircle Q A B ->
  exists Z, Bet P Z Q /\ OnCircle Z A B.
    
```

2 Liens entre les axiomes de continuité




FIGURE 5 – Aperçu des liens entre les axiomes de continuité

La figure 5 présente la hiérarchie des axiomes de continuité en géométrie neutre. Les ovales représentent chacun un axiome ou un groupe d’axiomes dont nous avons formalisé l’équivalence, à l’exception de l’ovale “Intégrité”, dont la définition dépend de la dimension, comme nous l’avons vu dans le paragraphe 1.2.2. Les implications que nous avons formalisées sont désignées par des flèches en trait

plein. Les flèches pointillées décrivent des implications connues dans la littérature [Str73] mais absentes de notre bibliothèque.

Les axiomes présentés et les preuves associées ont été formalisés en logique intuitionniste, à quelques exceptions près : les formalisations des implications Dedekind \Rightarrow Archimède et Segments emboîtés \Rightarrow Intégrité linéaire, ainsi que celle de l'équivalence des deux versions de Dedekind, recourent à la logique classique.

La plupart des implications sont strictes, mais nous n'avons pas transcrit les preuves d'indépendance. Ainsi, l'axiome d'Archimède n'est pas une conséquence de l'axiome d'Aristote [Gre88, Gre10]. Cependant, l'axiome d'Aristote peut quant à lui être dérivé de l'axiome de Greenberg dans l'hypothèse où la somme des angles d'un triangle fait un angle plat [BGNS17]; la question de savoir si c'est toujours le cas sans cette hypothèse reste ouverte.

2.1 La conjonction Segments emboîtés + Archimède implique Dedekind

Nous avons formalisé une preuve par dichotomie que la conjonction des axiomes des segments emboîtés et d'Archimède implique celui de Dedekind. En d'autres termes, notre preuve repose sur une suite de segments dont chacun est une des deux moitiés du précédent, et qui converge vers le point que l'on cherche à construire, à savoir le point de résolution d'une coupure de Dedekind. Nous devons donc associer à toute coupure deux prédicats décrivant les suites des deux extrémités de tels segments emboîtés. Pour cela, il suffit de s'assurer que chacun des segments traverse la coupure, c'est-à-dire que les deux sous-ensembles concernés sont représentés chacun par une des extrémités du segment. À chaque étape, on utilisera ce critère pour décider de définir un des segments emboîtés comme étant l'une ou l'autre moitié du précédent¹⁰ :

```

Inductive cX A C (Alpha Beta : Tpoint -> Prop) : nat -> Tpoint -> Prop :=
| cX_init : cX A C Alpha Beta O A
| cX_same : forall n X Y M, cX A C Alpha Beta n X -> cY A C Alpha Beta n Y ->
    Midpoint M X Y -> Beta M -> cX A C Alpha Beta (S n) X
| cX_other : forall n X Y M, cX A C Alpha Beta n X -> cY A C Alpha Beta n Y ->
    Midpoint M X Y -> Alpha M -> cX A C Alpha Beta (S n) M
with cY A C (Alpha Beta : Tpoint -> Prop) : nat -> Tpoint -> Prop :=
| cY_init : cY A C Alpha Beta O C
| cY_same : forall n X Y M, cX A C Alpha Beta n X -> cY A C Alpha Beta n Y ->
    Midpoint M X Y -> Alpha M -> cY A C Alpha Beta (S n) Y
| cY_other : forall n X Y M, cX A C Alpha Beta n X -> cY A C Alpha Beta n Y ->
    Midpoint M X Y -> Beta M -> cY A C Alpha Beta (S n) M.

```

Ici, *Alpha* et *Beta* désignent les deux prédicats caractérisant la coupure de Dedekind, tandis que *A* est l'origine de la demi-droite coupée et *C* un témoin de *Beta*. Notre définition mutuellement inductive permet de prouver, en raisonnant par induction mutuelle, que les prédicats *cX A C Alpha Beta* et *cY A C Alpha Beta* décrivent des segments emboîtés. L'axiome des segments emboîtés garantit donc l'existence d'un point *B* appartenant à chacun des segments. Comme chaque segment est deux fois plus petit que le précédent, l'axiome d'Archimède permet de prouver qu'ils convergent vers le point *B*, dont on démontre alors qu'il résout la coupure.

2.2 Dedekind implique l'intégrité linéaire

L'axiome d'intégrité linéaire est une conséquence de l'axiome de Dedekind¹¹. Pour formaliser la preuve de cette implication entre des axiomes géométriques, nous nous sommes inspirés de la preuve

10. La définition des moitiés d'un segment repose sur l'existence de son milieu, qui a été démontrée par Gupta [Gup65].

11. Cette implication n'est pas directement représentée sur la figure 5, car le lecteur peut la déduire des autres implications.

de James Forsythe Hall concernant les énoncés analogues dans les corps ordonnés [Hall1].

Cette preuve repose sur le fait que l'image de toute extension f dans un corps archimédien est dense dans ce corps, c'est-à-dire qu'entre deux éléments distincts A et B du corps d'arrivée, il existe toujours un élément X tel que $f(X)$ est situé strictement entre A et B . Notre traduction de cette propriété en termes géométriques est la suivante :

```
Lemma extension_image_density : forall {Tm: Tarski_neutral_dimensionless}
  {Tm2 : Tarski_neutral_dimensionless_with_decidable_point_equality Tm}
  P Q (f : @Tpoint Tn -> @Tpoint Tm),
  @archimedes_axiom Tm ->
  line_extension f P Q ->
  forall A B, Col (f P) (f Q) A -> Col (f P) (f Q) B -> A <> B ->
  exists X, Col P Q X /\ Bet A (f X) B /\ f X <> A /\ f X <> B.
```

Cependant, la preuve originale établit cette propriété en utilisant explicitement la structure de corps : elle s'appuie sur la densité de \mathbb{Q} dans tout corps archimédien. Or, les axiomes de la géométrie neutre ne nous permettent pas de définir de structure de corps au sein d'une droite de notre espace : l'axiome des parallèles (caractérisant la géométrie euclidienne) est requis.

Nous avons donc dû trouver un autre moyen de prouver cette propriété. Nous l'avons fait en introduisant principalement deux lemmes intermédiaires. Le premier établit que pour toute extension de droite f , tous points A et B dans la droite de départ, et tout entier n , il existe un point C tel que $f(A)f(C) = f(A)f(B) \times n$ (il s'agit en fait du point C tel que $AC = AB \times n$). En effet, multiplier un segment par un entier n utilise pas de structure de corps, mais le prédicat `Grad` que nous avons défini dans le paragraphe 1.2.1 :

```
Lemma extension_grad : forall {Tm: Tarski_neutral_dimensionless}
  {Tm2 : Tarski_neutral_dimensionless_with_decidable_point_equality Tm}
  P Q (f : @Tpoint Tn -> @Tpoint Tm),
  line_extension f P Q ->
  forall A B X, Col P Q A -> Col P Q B -> Grad (f A) (f B) X ->
  exists C, Col P Q C /\ Grad A B C /\ f C = X.
```

Comme l'axiome d'intégrité comporte un espace archimédien, il était prévisible que notre démonstration fasse appel au prédicat `Grad`, sur lequel repose notre définition de l'axiome d'Archimède. Notre deuxième lemme intermédiaire utilise une de ses variantes : `GradExp A B C` signifie que le segment AC est le produit du segment AB par une puissance de 2. Cela revient à dire que le segment AB est le quotient de AC par un entier de la forme 2^n , ou encore qu'il existe un entier n tel que le segment AB est obtenu après n bisections successives de AC ¹². Ici, il nous permet de formuler le fait que pour toute extension de droite f , tous points A et B dans la droite de départ, et tout entier naturel n , il existe un point C tel que $f(A)f(C) = f(A)f(B)/2^n$ (il s'agit en fait du point C tel que $AC = AB/2^n$)¹³ :

```
Inductive GradExp : Tpoint -> Tpoint -> Tpoint -> Prop :=
| gradexp_init : forall A B, GradExp A B B
| gradexp_stab : forall A B C C',
  GradExp A B C ->
  Bet A C C' -> Cong A C C C' ->
  GradExp A B C'.
```

12. On peut remarquer que ce prédicat est utilisé pour raisonner par dichotomie dans la formalisation de la sous-partie 2.1.

13. Remarquons qu'entre la définition du prédicat `GradExp` et celle de notre lemme, nous avons interverti les rôles des points B et C , afin de valoriser le fait que le point C est défini en fonction de A et B .

```

Lemma extension_gradexp : forall {Tm: Tarski_neutral_dimensionless}
  {Tm2 : Tarski_neutral_dimensionless_with_decidable_point_equality Tm}
  P Q (f : @Tpoint Tn -> @Tpoint Tm),
  line_extension f P Q ->
  forall A B X, Col P Q A -> Col P Q B -> GradExp (f A) X (f B) ->
  exists C, Col P Q C /\ GradExp A C B /\ f C = X.

```

La combinaison de ces deux lemmes intermédiaires nous permet d'affirmer que pour toute extension de droite f et tous points A et B de la droite de départ, l'image de f comprend tout point X de la droite d'arrivée issu d'une construction de la forme $f(A)X = f(A)f(B) \times n/2^m$, où n et m désignent des entiers naturels. Si la droite d'arrivée est archimédienne, on peut en déduire que l'image de f est dense dans cette droite, ce qui constitue le principal argument que l'axiome de Dedekind implique celui d'intégrité linéaire :

Démonstration. Étant donné une extension f d'une droite PQ et un point A sur la droite $f(P)f(Q)$, nous devons montrer qu'il existe un point B tel que $f(B) = A$. Sans perte de généralité on peut supposer que A est différent de $f(P)$. Soit A_1 le symétrique de $f(P)$ par rapport à A . Le lemme de densité de l'image nous permet de construire un point R tel que $f(R)$ est situé strictement entre A et A_1 . Le point A est donc situé entre $f(P)$ et $f(R)$. On considère ensuite la coupure de la demi-droite $]PR$ distinguant les points dont l'image par f est située strictement entre $f(P)$ et A et les points dont l'image est située sur la demi-droite $[Af(R)$. L'axiome de Dedekind nous permet d'obtenir un point B résolvant cette coupure. Si $f(B)$ était différent de A , on obtiendrait par densité un point X dont l'image se situerait strictement entre A et $f(B)$, ce qui contredirait la définition de B . On a donc $f(B) = A$. \square

Conclusion

Nos précédents résultats sur la continuité avaient contribué à la classification de différentes versions du postulat des parallèles [BGNS17]. Dans cet article, nous avons étendu cette étude pour y intégrer des axiomes de continuité historiquement importants et reconstruire une hiérarchie dans les axiomes de continuité. Nous avons aussi exhibé un chemin permettant de remonter à l'axiome le plus fort, à savoir celui de Dedekind.

Suite à nos précédents travaux, la bibliothèque *GeoCoq* permettait de choisir parmi 34 axiomes des parallèles. Dorénavant, l'utilisateur peut aussi choisir le ou les axiomes de continuité les plus adaptés à ses besoins. Ainsi, l'utilisateur dérouté par l'axiome d'intégrité pourra lui préférer l'axiome des segments emboîtés de Cantor. Outre sa définition géométriquement intuitive, l'axiome des segments emboîtés présente à la fois l'avantage d'être indépendant de l'axiome d'Archimède et de suffire à obtenir la complétude en présence d'Archimède.

L'originalité de nos démonstrations réside dans le fait qu'elles s'inscrivent dans la géométrie neutre de Tarski et que nous travaillons majoritairement en logique intuitionniste.

Ce travail peut avoir plusieurs extensions pouvant se recouper entre elles. D'abord, certains des axiomes de continuité pour les corps ordonnés listés dans [Hal11] (par exemple la version indénombrable de l'axiome des segments emboîtés) pourraient être adaptés à la géométrie et intégrés à notre étude. Ensuite, nous pourrions nous restreindre à la géométrie euclidienne et utiliser son arithmétisation [BBN18] pour prouver l'équivalence entre certains axiomes de continuité géométriques et leurs traductions respectives au sein des corps ordonnés. Enfin, il serait intéressant de clarifier notre hiérarchie en formalisant les résultats d'indépendance ou d'implication entre nos différents axiomes.

Disponibilité Le développement Coq décrit dans cet article est disponible dans la bibliothèque *GeoCoq* à partir de la version 2.4.0 : <http://geocoq.github.io/GeoCoq/>

Références

- [BBN16] Gabriel Braun, Pierre Boutry, and Julien Narboux. From Hilbert to Tarski. In *Eleventh International Workshop on Automated Deduction in Geometry*, Proceedings of ADG 2016, page 19, Strasbourg, France, June 2016.
- [BBN18] Pierre Boutry, Gabriel Braun, and Julien Narboux. Formalization of the Arithmetization of Euclidean Plane Geometry and Applications. *Journal of Symbolic Computation*, page 23, 2018. to appear.
- [BGNS17] Pierre Boutry, Charly Gries, Julien Narboux, and Pascal Schreck. Parallel postulates and continuity axioms : a mechanized study in intuitionistic logic using Coq. *Journal of Automated Reasoning*, page 68, 2017. online first.
- [BN12] Gabriel Braun and Julien Narboux. From Tarski to Hilbert. In Tetsuo Ida and Jacques Fleuriot, editors, *Post-proceedings of Automated Deduction in Geometry 2012*, volume 7993 of *LNCS*, pages 89–109, Edinburgh, United Kingdom, September 2012. Springer.
- [Can74] Georg Cantor. Ueber eine eigenschaft des inbegriffs aller reellen algebraischen zahlen. *Journal für die reine und angewandte Mathematik*, 77 :258–262, 1874.
- [Deh00] M. Dehn. Die Legendre’schen Sätze über die Winkelsumme im Dreieck. *Mathematische Annalen*, 53(3) :404–439, 1900.
- [Dup10] Jean Duprat. Fondements de géométrie euclidienne. 2010.
- [Ehr97] Philip Ehrlich. From Completeness to Archimedean Completeness : An Essay in the Foundations of Euclidean Geometry. *A Symposium on David Hilbert*, 110 :57–76, 1997.
- [GBN16] Charly Gries, Pierre Boutry, and Julien Narboux. Somme des angles d’un triangle et unicité de la parallèle : une preuve d’équivalence formalisée en Coq. In *Les vingt-septièmes Journées Francophones des Langages Applicatifs (JFLA 2016)*, Actes des Vingt-septièmes Journées Francophones des Langages Applicatifs (JFLA 2016), page 15, Saint Malo, France, January 2016. Jade Algave and Julien Signoles.
- [Gio13] Eduardo Giovannini. Completitud y continuidad en Fundamentos de la Geometría de Hilbert : acerca del Vollständigkeitsaxiom. *THEORIA. An International Journal for Theory, History and Foundations of Science*, 28(1) :139–163, 2013.
- [Gre88] Marvin Jay Greenberg. Aristotle’s axiom in the foundations of geometry. *Journal of Geometry*, 33(1) :53–57, 1988.
- [Gre10] Marvin J. Greenberg. Old and New Results in the Foundations of Elementary Plane Euclidean and Non-Euclidean Geometries. *The American Mathematical Monthly*, 117(3) :198–219, March 2010.
- [Gup65] Haragauri Narayan Gupta. *Contributions to the axiomatic foundations of geometry*. PhD thesis, University of California, Berkley, 1965.
- [Hal11] James Forsythe Hall. Completeness of ordered fields. *arXiv preprint arXiv :1101.5652*, 2011.
- [Har00] Robin Hartshorne. *Geometry : Euclid and beyond*. Undergraduate texts in mathematics. Springer, 2000.
- [Hil99] David Hilbert. *Grundlagen der Geometrie*. Leipzig, 1899.
- [Hil77] David Hilbert. *Foundations of Geometry*. Paul Bernays, open court publishing edition, 1977. 10th Revised edition.
- [Mar98] G. E. Martin. *The Foundations of Geometry and the Non-Euclidean Plane*. Undergraduate Texts in Mathematics. Springer, 1998.
- [Pas82] Moritz Pasch. *Vorlesungen über neuere Geometrie*. Teubner, Leipzig, 1882.
- [Rot14] Franz Rothe. *Several Topics from Geometry*. unpublished, 2014.
- [SST83] Wolfram Schwabhäuser, Wanda Szmielew, and Alfred Tarski. *Metamathematische Methoden in der Geometrie*. Springer-Verlag, Berlin, 1983.
- [Str73] J. Strommer. Über die Kreisaxiome. *Periodica Mathematica Hungarica*, 4 :3–16, 1973.
- [Tar51] Alfred Tarski. *A decision method for elementary algebra and geometry*. University of California Press, 1951.
- [Web38] Werner Weber. Über die Widersprüche in gewissen linearen Vollständigkeitsaxiomen der Geometrie. *Sitzungsber. Preuß. Akad. Wiss., Phys.-Math. Kl.*, 1938 :376–382, 1938.